

# Verification of Selection and Heap Sort Using Locales

Danijela Petrović

May 26, 2024

## Abstract

Stepwise program refinement techniques can be used to simplify program verification. Programs are better understood since their main properties are clearly stated, and verification of rather complex algorithms is reduced to proving simple statements connecting successive program specifications. Additionally, it is easy to analyze similar algorithms and to compare their properties within a single formalization. Usually, formal analysis is not done in educational setting due to complexity of verification and a lack of tools and procedures to make comparison easy. Verification of an algorithm should not only give correctness proof, but also better understanding of an algorithm. If the verification is based on small step program refinement, it can become simple enough to be demonstrated within the university-level computer science curriculum. In this paper we demonstrate this and give a formal analysis of two well known algorithms (Selection Sort and Heap Sort) using proof assistant Isabelle/HOL and program refinement techniques.

## Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Locale Sort</b>	<b>4</b>
<b>3</b>	<b>Defining data structure and key function <code>remove_max</code></b>	<b>5</b>
3.1	Describing data structure . . . . .	5
3.2	Function <code>remove_max</code> . . . . .	6
<b>4</b>	<b>Verification of functional Selection Sort</b>	<b>8</b>
4.1	Defining data structure . . . . .	9
4.2	Defining function <code>remove_max</code> . . . . .	9

<b>5</b>	<b>Verification of Heap Sort</b>	<b>10</b>
5.1	Defining tree and properties of heap . . . . .	10
<b>6</b>	<b>Verification of Functional Heap Sort</b>	<b>12</b>
<b>7</b>	<b>Verification of Imperative Heap Sort</b>	<b>13</b>
<b>8</b>	<b>Related work</b>	<b>16</b>
<b>9</b>	<b>Conclusions and Further Work</b>	<b>17</b>

## 1 Introduction

**Using program verification within computer science education.** Program verification is usually considered to be too hard and long process that acquires good mathematical background. A verification of a program is performed using mathematical logic. Having the specification of an algorithm inside the logic, its correctness can be proved again by using the standard mathematical apparatus (mainly induction and equational reasoning). These proofs are commonly complex and the reader must have some knowledge about mathematical logic. The reader must be familiar with notions such as satisfiability, validity, logical consequence, etc. Any misunderstanding leads into a loss of accuracy of the verification. These formalizations have common disadvantage, they are too complex to be understood by students, and this discourage students most of the time. Therefore, programmers and their educators rather use traditional (usually trial-and-error) methods.

However, many authors claim that nowadays education lacks the formal approach and it is clear why many advocate in using proof assistants[9]. This is also the case with computer science education. Students are presented many algorithms, but without formal analysis, often omitting to mention when algorithm would not work properly. Frequently, the center of a study is implementation of an algorithm whereas understanding of its structure and its properties is put aside. Software verification can bring more formal approach into teaching of algorithms and can have some advantages over traditional teaching methods.

- Verification helps to point out what are the requirements and conditions that an algorithm satisfies (pre-conditions, post-conditions and invariant conditions) and then to apply this knowledge during programming. This would help both students and educators to better understand input and output specification and the relations between them.
- Though program works in general case, it can happen that it does not work for some inputs and students must be able to detect these

situations and to create software that works properly for all inputs.

- It is suitable to separate abstract algorithm from its specific implementation. Students can compare properties of different implementations of the same algorithms, to see the benefits of one approach or another. Also, it is possible to compare different algorithms for same purpose (for example, for searching element, sorting, etc.) and this could help in overall understanding of algorithm construction techniques.

Therefore, lessons learned from formal verification of an algorithm can improve someones style of programming.

**Modularity and refinement.** The most used languages today are those who can easily be compiled into efficient code. Using heuristics and different data types makes code more complex and seems to novices like perplex mixture of many new notions, definitions, concepts. These techniques and methods in programming makes programs more efficient but are rather hard to be intuitively understood. On the other hand highly accepted principle in nowadays programming is modularity. Adhering to this principle enables programmer to easily maintain the code.

The best way to apply modularity on program verification and to make verification flexible enough to add new capabilities to the program keeping current verification intact is *program refinement*. Program refinement is the verifiable transformation of an abstract (high-level) formal specification into a concrete (low-level) executable program. It starts from the abstract level, describing only the requirements for input and output. Implementation is obtained at the end of the verification process (often by means of code generation [5]). Stepwise refinement allows this process to be done in stages. There are many benefits of using refinement techniques in verification.

- It gives a better understanding of programs that are verified.
- The algorithm can be analyzed and understood on different level of abstraction.
- It is possible to verify different implementations for some part of the program, discussing the benefits of one approach or another.
- It can be easily proved that these different implementation share some same properties which are proved before splitting into two directions.
- It is easy to maintain the code and the verification. Usually, whenever the implementation of the program changes, the correctness proofs must be adapted to these changes, and if refinement is used, it is not necessary to rewrite entire verification, just add or change small part of it.

- Using refinement approach makes algorithm suitable for a case study in teaching. Properties and specifications of the program are clearly stated and it helps teachers and students better to teach or understand them.

We claim that the full potential of refinement comes only when it is applied stepwise, and in many small steps. If the program is refined in many steps, and data structures and algorithms are introduced one-by-one, then proving the correctness between the successive specifications becomes easy. Abstracting and separating each algorithmic idea and each data-structure that is used to give an efficient implementation of an algorithm is very important task in programmer education.

As an example of using small step refinement, in this paper we analyze two widely known algorithms, Selection Sort and Heap Sort. There are many reasons why we decided to use them.

- They are largely studied in different contexts and they are studied in almost all computer science curricula.
- They belong to the same family of algorithms and they are good example for illustrating the refinement techniques. They are a nice example of how one can improve on a same idea by introducing more efficient underlying data-structures and more efficient algorithms.
- Their implementation uses different programming constructs: loops (or recursion), arrays (or lists), trees, etc. We show how to analyze all these constructs in a formal setting.

There are many formalizations of sorting algorithms that are done both automatically or interactively and they undoubtedly proved that these algorithms are correct. In this paper we are giving a new approach in their verification, that insists on formally analyzing connections between them, instead of only proving their correctness (which has been well established many times). Our central motivation is that these connections contribute to deeper algorithm understanding much more than separate verification of each algorithm.

## 2 Locale Sort

```
theory Sort
imports Main
  HOL-Library.Multiset
begin
```

First, we start from the definition of sorting algorithm. *What are the basic properties that any sorting algorithm must satisfy?* There are two basic features any sorting algorithm must satisfy:

- The elements of sorted array must be in some order, e.g. ascending or descending order. In this paper we are sorting in ascending order.

$$\text{sorted} (\text{sort } l)$$

- The algorithm does not change or delete elements of the given array, e.g. the sorted array is the permutation of the input array.

$$\text{sort } l <\sim\sim> l$$

```

locale Sort =
  fixes sort :: 'a::linorder list  $\Rightarrow$  'a list
  assumes sorted: sorted (sort l)
  assumes permutation: mset (sort l) = mset l

```

**end**

### 3 Defining data structure and key function remove\_\_max

```

theory RemoveMax
imports Sort
begin

```

#### 3.1 Describing data structure

We have already said that we are going to formalize heap and selection sort and to show connections between these two sorts. However, one can immediately notice that selection sort is using list and heap sort is using heap during its work. It would be very difficult to show equivalency between these two sorts if it is continued straightforward and independently proved that they satisfy conditions of locale `Sort`. They work with different objects. Much better thing to do is to stay on the abstract level and to add the new locale, one that describes characteristics of both list and heap.

```

locale Collection =
  fixes empty :: 'b
  — — Represents empty element of the object (for example, for list it is [])
  fixes is-empty :: 'b  $\Rightarrow$  bool
  — — Function that checks weather the object is empty or not
  fixes of-list :: 'a list  $\Rightarrow$  'b
  — — Function transforms given list to desired object (for example, for heap sort,
function of_list transforms list to heap)
  fixes multiset :: 'b  $\Rightarrow$  'a multiset
  — — Function makes a multiset from the given object. A multiset is a collection
without order.
  assumes is-empty-inj: is-empty e  $\Longrightarrow$  e = empty

```

— — It must be assured that the empty element is *empty*  
**assumes** *is-empty-empty*: *is-empty empty*  
— — Must be satisfied that function *is\_empty* returns true for element *empty*  
**assumes** *multiset-empty*: *multiset empty = {#}*  
— — Multiset of an empty object is empty multiset.  
**assumes** *multiset-of-list*: *multiset (of-list i) = mset i*  
— — Multiset of an object gained by applying function *of\_list* must be the same as the multiset of the list. This, practically, means that function *of\_list* does not delete or change elements of the starting list.  
**begin**  
**lemma** *is-empty-as-list*: *is-empty e  $\implies$  multiset e = {#}*  
  <proof>  
  
**definition** *set* :: 'b  $\Rightarrow$  'a *set* **where**  
  [*simp*]: *set l = set-mset (multiset l)*  
**end**

### 3.2 Function `remove_max`

We wanted to emphasize that algorithms are same. Due to the complexity of the implementation it usually happens that simple properties are omitted, such as the connection between these two sorting algorithms. This is a key feature that should be presented to students in order to understand these algorithms. It is not unknown that students usually prefer selection sort for its simplicity whereas avoid heap sort for its complexity. However, if we can present them as the algorithms that are same they may hesitate less in using the heap sort. This is why the refinement is important. Using this technique we were able to notice these characteristics. Separate verification would not bring anything new. Being on the abstract level does not only simplify the verifications, but also helps us to notice and to show students important features. Even further, we can prove them formally and completely justify our observation.

**locale** *RemoveMax* = *Collection empty is-empty of-list multiset* **for**  
  *empty* :: 'b **and**  
  *is-empty* :: 'b  $\Rightarrow$  *bool* **and**  
  *of-list* :: 'a::linorder *list*  $\Rightarrow$  'b **and**  
  *multiset* :: 'b  $\Rightarrow$  'a::linorder *multiset* +  
  **fixes** *remove-max* :: 'b  $\Rightarrow$  'a  $\times$  'b  
  — — Function that removes maximum element from the object of type 'b. It returns maximum element and the object without that maximum element.  
  **fixes** *inv* :: 'b  $\Rightarrow$  *bool*  
  — — It checks weather the object is in required condition. For example, if we expect to work with heap it checks weather the object is heap. This is called *invariant condition*  
  **assumes** *of-list-inv*: *inv (of-list x)*  
  — — This condition assures that function *of\_list* made a object with desired property.

**assumes** *remove-max-max*:

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l \rrbracket \implies m = \text{Max } (\text{set } l)$

— — First parameter of the return value of the function *remove\_max* is the maximum element

**assumes** *remove-max-multiset*:

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l \rrbracket \implies$   
 $\text{add-mset } m \text{ (multiset } l') = \text{multiset } l$

— — Condition for multiset, ensures that nothing new is added or nothing is lost after applying *remove\_max* function.

**assumes** *remove-max-inv*:

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l \rrbracket \implies \text{inv } l'$

— — Ensures that invariant condition is true after removing maximum element.

Invariant condition must be true in each step of sorting algorithm, for example if we are sorting using heap than in each iteration we must have heap and function *remove\_max* must not change that.

**begin**

**lemma** *remove-max-multiset-size*:

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l \rrbracket \implies$   
 $\text{size } (\text{multiset } l) > \text{size } (\text{multiset } l')$

*<proof>*

**lemma** *remove-max-set*:

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l \rrbracket \implies$   
 $\text{set } l' \cup \{m\} = \text{set } l$

*<proof>*

As it is said before in each iteration invariant condition must be satisfied, so the *inv l* is always true, e.g. before and after execution of any function. This is also the reason why sort function must be defined as partial. This function parameters stay the same in each step of iteration – list stays list, and heap stays heap. As we said before, in Isabelle/HOL we can only define total function, but there is a mechanism that enables total function to appear as partial one:

**partial-function** (*tailrec*) *ssort'* **where**

*ssort' l sl =*  
  (*if is-empty l then*  
    *sl*  
  else  
    *let*  
      (*m, l' = remove-max l*  
    in  
      *ssort' l' (m # sl)*)

**declare** *ssort'.simps*[code]

**definition** *ssort* :: 'a list  $\Rightarrow$  'a list **where**

*ssort l = ssort' (of-list l)* []

**inductive** *ssort'-dom* **where**

step:  $\llbracket \bigwedge m l'. \llbracket \neg \text{is-empty } l; (m, l') = \text{remove-max } l \rrbracket \implies \text{ssort}'\text{-dom } (l', m \# sl) \rrbracket \implies \text{ssort}'\text{-dom } (l, sl)$

**lemma** *ssort'-termination*:

**assumes** *inv (fst p)*

**shows** *ssort'-dom p*

*<proof>*

**lemma** *ssort'Induct*:

**assumes** *inv l P l sl*

$\bigwedge l sl m l'$ .

$\llbracket \neg \text{is-empty } l; \text{inv } l; (m, l') = \text{remove-max } l; P l sl \rrbracket \implies P l' (m \# sl)$

**shows** *P empty (ssort' l sl)*

*<proof>*

**lemma** *mset-ssort'*:

**assumes** *inv l*

**shows** *mset (ssort' l sl) = multiset l + mset sl*

*<proof>*

**lemma** *sorted-ssort'*:

**assumes** *inv l sorted sl  $\wedge (\forall x \in \text{set } l. (\forall y \in \text{List.set } sl. x \leq y))$*

**shows** *sorted (ssort' l sl)*

*<proof>*

**lemma** *sorted-ssort: sorted (ssort i)*

*<proof>*

**lemma** *permutation-ssort: mset (ssort l) = mset l*

*<proof>*

**end**

Using assumptions given in the definitions of the locales *Collection* and *RemoveMax* for the functions *multiset*, *is\_empty*, *of\_list* and *remove\_max* it is no difficulty to show:

**sublocale** *RemoveMax < Sort ssort*

*<proof>*

**end**

## 4 Verification of functional Selection Sort

**theory** *SelectionSort-Functional*

**imports** *RemoveMax*

**begin**



## 4.1 Defining data structure

Selection sort works with list and that is the reason why *Collection* should be interpreted as list.

**interpretation** *Collection* []  $\lambda l. l = []$  *id mset*  
{proof}

## 4.2 Defining function remove\_max

The following is definition of *remove\_max* function. The idea is very well known – assume that the maximum element is the first one and then compare with each element of the list. Function *f* is one step in iteration, it compares current maximum *m* with one element *x*, if it is bigger then *m* stays current maximum and *x* is added in the resulting list, otherwise *x* is current maximum and *m* is added in the resulting list.

**fun** *f* **where**  $f (m, l) x = (if\ x \geq\ m\ then\ (x, m\#\!l)\ else\ (m, x\#\!l))$

**definition** *remove-max* **where**  
 $remove-max\ l = foldl\ f\ (hd\ l, [])\ (tl\ l)$

**lemma** *max-Max-commute*:  
 $finite\ A \implies max\ (Max\ (insert\ m\ A))\ x = max\ m\ (Max\ (insert\ x\ A))$   
{proof}

The function really returned the maximum value.

**lemma** *remove-max-max-lemma*:  
**shows**  $fst\ (foldl\ f\ (m, t)\ l) = Max\ (set\ (m\ \#\ l))$   
{proof}

**lemma** *remove-max-max*:  
**assumes**  $l \neq []$   $(m, l') = remove-max\ l$   
**shows**  $m = Max\ (set\ l)$   
{proof}

Nothing new is added in the list and nothing is deleted from the list except the maximum element.

**lemma** *remove-max-mset-lemma*:  
**assumes**  $(m, l') = foldl\ f\ (m', t')\ l$   
**shows**  $mset\ (m\ \#\ l') = mset\ (m' \#\ t' \ @\ l)$   
{proof}

**lemma** *remove-max-mset*:  
**assumes**  $l \neq []$   $(m, l') = remove-max\ l$   
**shows**  $add-mset\ m\ (mset\ l') = mset\ l$   
{proof}

**definition** *ssf-ssort'* **where**

[simp, code del]:  $ssf\text{-}ssort' = RemoveMax.ssort' (\lambda l. l = []) \text{ remove-max}$   
**definition**  $ssf\text{-}ssort$  **where**  
 [simp, code del]:  $ssf\text{-}ssort = RemoveMax.ssort (\lambda l. l = []) \text{ id remove-max}$

**interpretation**  $SSRemoveMax$ :

$RemoveMax [] \lambda l. l = [] \text{ id mset remove-max } \lambda -. \text{ True}$   
**rewrites**  
 $RemoveMax.ssort' (\lambda l. l = []) \text{ remove-max} = ssf\text{-}ssort'$  **and**  
 $RemoveMax.ssort (\lambda l. l = []) \text{ id remove-max} = ssf\text{-}ssort$   
 <proof>

**end**

## 5 Verification of Heap Sort

**theory**  $Heap$   
**imports**  $RemoveMax$   
**begin**

### 5.1 Defining tree and properties of heap

**datatype**  $'a \text{ Tree} = E \mid T 'a 'a \text{ Tree} 'a \text{ Tree}$

With  $E$  is represented empty tree and with  $T 'a 'a \text{ Tree} 'a \text{ Tree}$  is represented a node whose root element is of type  $'a$  and its left and right branch is also a tree of type  $'a$ .

**primrec**  $size :: 'a \text{ Tree} \Rightarrow nat$  **where**  
 $size E = 0$   
 $| size (T v l r) = 1 + size l + size r$

Definition of the function that makes a multiset from the given tree:

**primrec**  $multiset$  **where**  
 $multiset E = \{\#\}$   
 $| multiset (T v l r) = multiset l + \{\#v\#\} + multiset r$

**primrec**  $val$  **where**  
 $val (T v -) = v$

Definition of the function that has the value  $True$  if the tree is heap, otherwise it is  $False$ :

**fun**  $is\text{-}heap :: 'a::linorder \text{ Tree} \Rightarrow bool$  **where**  
 $is\text{-}heap E = True$   
 $| is\text{-}heap (T v E E) = True$   
 $| is\text{-}heap (T v E r) = (v \geq val r \wedge is\text{-}heap r)$   
 $| is\text{-}heap (T v l E) = (v \geq val l \wedge is\text{-}heap l)$   
 $| is\text{-}heap (T v l r) = (v \geq val r \wedge is\text{-}heap r \wedge v \geq val l \wedge is\text{-}heap l)$

**lemma**  $heap\text{-}top\text{-}geq$ :

**assumes**  $a \in \# \text{ multiset } t \text{ is-heap } t$   
**shows**  $\text{val } t \geq a$   
 <proof>

**lemma** *heap-top-max*:  
**assumes**  $t \neq E \text{ is-heap } t$   
**shows**  $\text{val } t = \text{Max-mset } (\text{multiset } t)$   
 <proof>

The next step is to define function *remove\_max*, but the question is weather implementation of *remove\_max* depends on implementation of the functions *is\_heap* and *multiset*. The answer is negative. This suggests that another step of refinement could be added before definition of function *remove\_max*. Additionally, there are other reasons why this should be done, for example, function *remove\_max* could be implemented in functional or in imperative manner.

**locale** *Heap* = *Collection empty is-empty of-list multiset for*  
*empty* :: 'b **and**  
*is-empty* :: 'b  $\Rightarrow$  bool **and**  
*of-list* :: 'a::linorder list  $\Rightarrow$  'b **and**  
*multiset* :: 'b  $\Rightarrow$  'a::linorder multiset +  
**fixes** *as-tree* :: 'b  $\Rightarrow$  'a::linorder Tree

— This function is not very important, but it is needed in order to avoid problems with types and to detect that observed object is a tree.

**fixes** *remove-max* :: 'b  $\Rightarrow$  'a  $\times$  'b  
**assumes** *multiset*:  $\text{multiset } l = \text{Heap.multiset } (\text{as-tree } l)$   
**assumes** *is-heap-of-list*:  $\text{is-heap } (\text{as-tree } (\text{of-list } i))$   
**assumes** *as-tree-empty*:  $\text{as-tree } t = E \iff \text{is-empty } t$   
**assumes** *remove-max-multiset'*:  
 $\llbracket \neg \text{is-empty } l; (m, l') = \text{remove-max } l \rrbracket \implies \text{add-mset } m (\text{multiset } l') = \text{multiset } l$   
**assumes** *remove-max-is-heap*:  
 $\llbracket \neg \text{is-empty } l; \text{is-heap } (\text{as-tree } l); (m, l') = \text{remove-max } l \rrbracket \implies$   
 $\text{is-heap } (\text{as-tree } l')$   
**assumes** *remove-max-val*:  
 $\llbracket \neg \text{is-empty } t; (m, t') = \text{remove-max } t \rrbracket \implies m = \text{val } (\text{as-tree } t)$

It is very easy to prove that locale *Heap* is sublocale of locale *RemoveMax*

**sublocale** *Heap* <  
*RemoveMax empty is-empty of-list multiset remove-max  $\lambda$  t. is-heap (as-tree t)*  
 <proof>

**primrec** *in-tree* **where**  
*in-tree*  $v \ E = \text{False}$   
 $\mid \text{in-tree } v \ (T \ v' \ l \ r) \iff v = v' \vee \text{in-tree } v \ l \vee \text{in-tree } v \ r$

**lemma** *is-heap-max*:  
**assumes**  $\text{in-tree } v \ t \text{ is-heap } t$   
**shows**  $\text{val } t \geq v$

*<proof>*

**end**

## 6 Verification of Functional Heap Sort

**theory** *HeapFunctional*

**imports** *Heap*

**begin**

As we said before, maximum element of the heap is its root. So, finding maximum element is not difficulty. But, this element should also be removed and remainder after deleting this element is two trees, left and right branch of original heap. Those branches are also heaps by the definition of the heap. To maintain consistency, branches should be combined into one tree that satisfies heap condition:

**function** *merge* :: *'a::linorder Tree*  $\Rightarrow$  *'a Tree*  $\Rightarrow$  *'a Tree* **where**

*merge t1 E = t1*

| *merge E t2 = t2*

| *merge (T v1 l1 r1) (T v2 l2 r2) =*

*(if v1  $\geq$  v2 then T v1 (merge l1 (T v2 l2 r2)) r1*

*else T v2 (merge l2 (T v1 l1 r1)) r2)*

*<proof>*

**termination**

*<proof>*

**lemma** *merge-val*:

*val(merge l r) = val l  $\vee$  val(merge l r) = val r*

*<proof>*

Function *merge* merges two heaps into one:

**lemma** *merge-heap-is-heap*:

**assumes** *is-heap l is-heap r*

**shows** *is-heap (merge l r)*

*<proof>*

**definition** *insert* :: *'a::linorder*  $\Rightarrow$  *'a Tree*  $\Rightarrow$  *'a Tree* **where**

*insert v t = merge t (T v E E)*

**primrec** *hs-of-list* **where**

*hs-of-list [] = E*

| *hs-of-list (v # l) = insert v (hs-of-list l)*

**definition** *hs-is-empty* **where**

[*simp*]: *hs-is-empty t  $\longleftrightarrow$  t = E*

Definition of function *remove\_max*:

**fun** *hs-remove-max*:: *'a::linorder Tree*  $\Rightarrow$  *'a  $\times$  'a Tree* **where**

$hs\text{-remove-max } (T\ v\ l\ r) = (v, \text{merge } l\ r)$

**lemma** *merge-multiset*:

$\text{multiset } l + \text{multiset } g = \text{multiset } (\text{merge } l\ g)$

*<proof>*

Proof that defined functions are interpretation of abstract functions from locale *Collection*:

**interpretation** *HS*: *Collection E hs-is-empty hs-of-list multiset*

*<proof>*

Proof that defined functions are interpretation of abstract functions from locale *Heap*:

**interpretation** *Heap E hs-is-empty hs-of-list multiset id hs-remove-max*

*<proof>*

**end**

## 7 Verification of Imperative Heap Sort

**theory** *HeapImperative*

**imports** *Heap*

**begin**

**primrec** *left* ::  $'a\ Tree \Rightarrow 'a\ Tree$  **where**

$\text{left } (T\ v\ l\ r) = l$

**abbreviation** *left-val* ::  $'a\ Tree \Rightarrow 'a$  **where**

$\text{left-val } t \equiv \text{val } (\text{left } t)$

**primrec** *right* ::  $'a\ Tree \Rightarrow 'a\ Tree$  **where**

$\text{right } (T\ v\ l\ r) = r$

**abbreviation** *right-val* ::  $'a\ Tree \Rightarrow 'a$  **where**

$\text{right-val } t \equiv \text{val } (\text{right } t)$

**abbreviation** *set-val* ::  $'a\ Tree \Rightarrow 'a \Rightarrow 'a\ Tree$  **where**

$\text{set-val } t\ x \equiv T\ x\ (\text{left } t)\ (\text{right } t)$

The first step is to implement function *siftDown*. If some node does not satisfy heap property, this function moves it down the heap until it does. For a node is checked weather it satisfies heap property or not. If it does nothing is changed. If it does not, value of the root node becomes a value of the larger child and the value of that child becomes the value of the root node. This is the reason this function is called **siftDown** – value of the node is places down in the heap. Now, the problem is that the child node may not satisfy the heap property and that is the reason why function **siftDown** is recursively applied.

```

fun siftDown :: 'a::linorder Tree ⇒ 'a Tree where
  siftDown E = E
| siftDown (T v E E) = T v E E
| siftDown (T v l E) =
  (if v ≥ val l then T v l E else T (val l) (siftDown (set-val l v)) E)
| siftDown (T v E r) =
  (if v ≥ val r then T v E r else T (val r) E (siftDown (set-val r v)))
| siftDown (T v l r) =
  (if val l ≥ val r then
    if v ≥ val l then T v l r else T (val l) (siftDown (set-val l v)) r
  else
    if v ≥ val r then T v l r else T (val r) l (siftDown (set-val r v)))

```

**lemma** *siftDown-Node*:  
**assumes**  $t = T v l r$   
**shows**  $\exists l' v' r'. \text{siftDown } t = T v' l' r' \wedge v' \geq v$   
*<proof>*

**lemma** *siftDown-in-tree*:  
**assumes**  $t \neq E$   
**shows** *in-tree* (val (siftDown t)) t  
*<proof>*

**lemma** *siftDown-in-tree-set*:  
**shows** *in-tree* v t  $\longleftrightarrow$  *in-tree* v (siftDown t)  
*<proof>*

**lemma** *siftDown-heap-is-heap*:  
**assumes** *is-heap* l *is-heap* r  $t = T v l r$   
**shows** *is-heap* (siftDown t)  
*<proof>*

Definition of the function *heapify* which makes a heap from any given binary tree.

```

primrec heapify where
  heapify E = E
| heapify (T v l r) = siftDown (T v (heapify l) (heapify r))

```

**lemma** *heapify-heap-is-heap*:  
*is-heap* (heapify t)  
*<proof>*

Definition of *removeLeaf* function. Function returns two values. The first one is the value of removed leaf element. The second returned value is tree without that leaf.

```

fun removeLeaf :: 'a::linorder Tree ⇒ 'a × 'a Tree where
  removeLeaf (T v E E) = (v, E)
| removeLeaf (T v l E) = (fst (removeLeaf l), T v (snd (removeLeaf l)) E)
| removeLeaf (T v E r) = (fst (removeLeaf r), T v E (snd (removeLeaf r)))

```

|  $removeLeaf (T v l r) = (fst (removeLeaf l), T v (snd (removeLeaf l)) r)$

Function  $of\_list\_tree$  makes a binary tree from any given list.

**primrec**  $of\_list\_tree:: 'a::linorder list \Rightarrow 'a Tree$  **where**  
 $of\_list\_tree [] = E$   
|  $of\_list\_tree (v \# tail) = T v (of\_list\_tree tail) E$

By applying  $heapify$  binary tree is transformed into heap.

**definition**  $hs\_of\_list$  **where**  
 $hs\_of\_list l = heapify (of\_list\_tree l)$

Definition of function  $hs\_remove\_max$ . As it is already well established, finding maximum is not a problem, since it is in the root element of the heap. The root element is replaced with leaf of the heap and that leaf is erased from its previous position. However, now the new root element may not satisfy heap property and that is the reason to apply function  $siftDown$ .

**definition**  $hs\_remove\_max:: 'a::linorder Tree \Rightarrow 'a \times 'a Tree$  **where**  
 $hs\_remove\_max t \equiv$   
 $(let v' = fst (removeLeaf t);$   
 $t' = snd (removeLeaf t) in$   
 $(if t' = E then (val t, E)$   
 $else (val t, siftDown (set-val t' v'))))$

**definition**  $hs\_is\_empty$  **where**  
 $[simp]: hs\_is\_empty t \longleftrightarrow t = E$

**lemma**  $siftDown\_multiset$ :  
 $multiset (siftDown t) = multiset t$   
 $\langle proof \rangle$

**lemma**  $mset\_list\_tree$ :  
 $multiset (of\_list\_tree l) = mset l$   
 $\langle proof \rangle$

**lemma**  $multiset\_heapify$ :  
 $multiset (heapify t) = multiset t$   
 $\langle proof \rangle$

**lemma**  $multiset\_heapify\_of\_list\_tree$ :  
 $multiset (heapify (of\_list\_tree l)) = mset l$   
 $\langle proof \rangle$

**lemma**  $removeLeaf\_val\_val$ :  
**assumes**  $snd (removeLeaf t) \neq E$   $t \neq E$   
**shows**  $val t = val (snd (removeLeaf t))$   
 $\langle proof \rangle$

**lemma** *removeLeaf-heap-is-heap*:  
**assumes** *is-heap t t*  $\neq E$   
**shows** *is-heap (snd (removeLeaf t))*  
 $\langle proof \rangle$

Defined functions satisfy conditions of locale *Collection* and thus represent interpretation of this locale.

**interpretation** *HS*: *Collection E hs-is-empty hs-of-list multiset*  
 $\langle proof \rangle$

**lemma** *removeLeaf-multiset*:  
**assumes**  $(v', t') = \text{removeLeaf } t \ t \neq E$   
**shows**  $\{\#v'\#\} + \text{multiset } t' = \text{multiset } t$   
 $\langle proof \rangle$

**lemma** *set-val-multiset*:  
**assumes**  $t \neq E$   
**shows**  $\text{multiset } (\text{set-val } t \ v') + \{\#\text{val } t\#\} = \{\#v'\#\} + \text{multiset } t$   
 $\langle proof \rangle$

**lemma** *hs-remove-max-multiset*:  
**assumes**  $(m, t') = \text{hs-remove-max } t \ t \neq E$   
**shows**  $\{\#m\#\} + \text{multiset } t' = \text{multiset } t$   
 $\langle proof \rangle$

Defined functions satisfy conditions of locale *Heap* and thus represent interpretation of this locale.

**interpretation** *Heap E hs-is-empty hs-of-list multiset id hs-remove-max*  
 $\langle proof \rangle$

**end**

## 8 Related work

To study sorting algorithms from a top down was proposed in [7]. All sorting algorithms are based on divide-and-conquer algorithm and all sorts are divided into two groups: *hard\_split/easy\_join* and *easy\_split/hard\_join*. Following this idea in [8], authors described sorting algorithms using object-oriented approach. They suggested that this approach could be used in computer science education and that presenting sorting algorithms from top down will help students to understand them better.

The paper [1] represent different recursion patterns — catamorphism, anamor-



phism, hylomorphism and paramorphisms. Selection, bubble, merge, heap and quick sort are expressed using these patterns of recursion and it is shown that there is a little freedom left in implementation level. Also, connection between different patterns are given and thus a conclusion about connection between sorting algorithms can be easily conducted. Furthermore, in the paper are generalized tree data types – list, binary trees and binary leaf trees.

Satisfiability procedures for working with arrays was proposed in paper “What is decidable about arrays?”[3]. This procedure is called  $SAT_A$  and can give an answer if two arrays are equal or if array is sorted and so on. Completeness and soundness for procedures are proved. There are, though, several cases when procedures are unsatisfiable. They also studied theory of maps. One of the application for these procedures is verification of sorting algorithms and they gave an example that insertion sort returns sorted array.

Tools for program verification are developed by different groups and with different results. Some of them are automated and some are half-automated. Ralph-Johan Back and Johannes Eriksson [6] developed SOCOS, tool for program verification based on invariant diagrams. SOCOS environment supports interactive and non-interactive checking of program correctness. For each program tree types of verification conditions are generated: consistency, completeness and termination conditions. They described invariant-based programming in SOCOS. In [2] this tool was used to verify heap sort algorithm.

There are many tools for Java program developers made to automatically prove program correctness. Krakatoa Modeling Language (KML) is described in [10] with example of sorting algorithms. Refinement is not supported in KML and any refinement property could not automatically be proved. The language KML is also not formally verified, but some parts are proved by Alt-Ergo, Simplify and Yices. The paper proposed some improvements for working with permutation and arrays in KML. Why/Krakatoa/Caduceus[4] is a tool for deductive program verification for Java and C. The approach is to use Krakatoa and Caduceus to translate Java/C programs into Why program. This language is suitable for program verification. The idea is to generate verification conditions based on weakest precondition calculus.

## 9 Conclusions and Further Work

In this paper we illustrated a proof management technology. The methodology that we use in this paper for the formalization is refinement: the formalization begins with a most basic specification, which is then refined

by introducing more advanced techniques, while preserving the correctness. This incremental approach proves to be a very natural approach in formalizing complex software systems. It simplifies understanding of the system and reduces the overall verification effort.

Modularity is very popular in nowadays imperative languages. This approach could be used for software verification and Isabelle/HOL locales provide means for modular reasoning. They support multiple inheritance and this means that locales can imitate connections between functions, procedures or objects. It is possible to establish some general properties of an algorithm or to compare these properties. So, it is possible to compare programs. And this is a great advantage in program verification, something that is not done very often. This could help in better understanding of an algorithm which is essential for computer science education. So apart from being able to formalize verification in easier manner, this approach gives us opportunity to compare different programs. This was showed on Selection and Heap sort example and the connection between these two sorts was easy to comprehend. The value of this approach is not so much in obtaining a nice implementation of some algorithm, but in unraveling its structure. This is very important for computer science education and this can help in better teaching and understanding of an algorithms.

Using experience from this formalization, we came to conclusion that the general principle for refinement in program verification should be: *divide program into small modules (functions, classes) and verify each modulo separately in order that corresponds to the order in entire program implementation*. Someone may argue that this principle was not followed in each step of formalization, for example when we implemented *Selection sort* or when we defined *is\_heap* and *multiset* in one step, but we feel that those function were simple and deviations in their implementations are minimal.

The next step is to formally verify all sorting algorithms and using refinement method to formally analyze and compare different sorting algorithms.

## References

- [1] L. Augusteijn. Sorting morphisms. In *3rd International Summer School on Advanced Functional Programming, volume 1608 of LNCS*, pages 1–27. Springer-Verlag, 1998.
- [2] R.-J. Back and J. Eriksson. Correct-by-construction programming in the socos environment. *CTP Components for Educational Software*, page 16, 2011.
- [3] A. R. Bradley, Z. Manna, and H. B. Sipma. What’s decidable about arrays? In *Proceedings of the 7th international conference on Verifica-*

- tion, Model Checking, and Abstract Interpretation*, VMCAI'06, pages 427–442, Berlin, Heidelberg, 2006. Springer-Verlag.
- [4] J. christophe Filliâtre and C. Marché. The why/krakatoa/caduceus platform for deductive program verification. In *In CAV 07*, pages 173–177, 2007.
  - [5] F. Haftmann and L. Bulwahn. Code generation from isabelle/hol theories, 2007.
  - [6] R. johan Back, J. Eriksson, and M. Myreen. Verifying invariant based programs in the socos environment. In *In Teaching Formal Methods: Practice and Experience (BCS Electronic Workshops in Computing). BCS-FACS*, 2006.
  - [7] S. M. Merritt. An inverted taxonomy of sorting algorithms. *Commun. ACM*, 28(1):96–99, Jan. 1985.
  - [8] D. Z. Nguyen and S. B. Wong. Design patterns for sorting. *SIGCSE Bull.*, 33(1):263–267, Feb. 2001.
  - [9] T. Nipkow. Teaching semantics with a proof assistant: No more lsd trip proofs. In *Verification, Model Checking, and Abstract Interpretation*, pages 24–38. Springer, 2012.
  - [10] E. Tushkanova, A. Giorgetti, and O. Kouchnarenko. Specifying and Proving a Sorting Algorithm. Research Report RR2009-03, LIFC - Laboratoire d'Informatique de l'Université de Franche-Comté, Oct. 2009. 35 pages.